

## 5.3 Informative Comments

This section provides a short description of each ABNF rule listed above. The related brief and long syntax rules are grouped together with the same description. Where the syntaxes are the same, the rule is listed once and preceded with the text "BS/LS". Where the brief and long syntaxes are different, both rules are listed separately and preceded with "BS" and "LS" respectively.

<b>BS/LS: expressionConstraint</b> = ws ( refinedExpressionConstraint / compoundExpressionConstraint / dottedExpressionConstraint / subExpressionConstraint ) ws	
	An expression constraint is either a refined expression constraint, a compound expression constraint, a dotted expression constraint, or a sub expression constraint.
<b>BS/LS: refinedExpressionConstraint</b> = subExpressionConstraint ws ":" ws eclRefinement	
	A refined expression constraint includes a subexpression constraint followed by a refinement.
<b>BS/LS: compoundExpressionConstraint</b> = conjunctionExpressionConstraint / disjunctionExpressionConstraint / exclusionExpressionConstraint	
	A compound expression constraint contains two or more expression constraints joined by either a conjunction, disjunction or exclusion. When potential ambiguity in binary operator precedence may occur, round brackets must be used to clearly disambiguate the order in which these operator are applied. Brackets are not required in expression constraints in which all binary operators are conjunctions, or all binary operators are disjunctions. Please note that unary operators (i.e. constraint operators and member of functions) are always applied before binary operators (i.e. conjunction, disjunction and exclusion).
<b>BS/LS: conjunctionExpressionConstraint</b> = subExpressionConstraint 1*(ws conjunction ws subExpressionConstraint)	
	A conjunction expression constraint combines two or more expression constraints with a conjunction ("and") operator. More than one conjunction may be used without brackets. However any compound expression constraint (using a different binary operator) that appears within a conjunction expression constraint must be enclosed by brackets.
<b>BS/LS: disjunctionExpressionConstraint</b> = subExpressionConstraint 1*(ws disjunction ws subExpressionConstraint)	
	A disjunction expression constraint combines two or more expression constraints with a disjunction ("or") operator. More than one disjunction may be used without brackets. However any compound expression constraint (using a different binary operator) that appears within a disjunction expression constraint must be enclosed by brackets.
<b>BS/LS: exclusionExpressionConstraint</b> = subExpressionConstraint ws exclusion ws subExpressionConstraint	
	An exclusion expression constraint combines two expression constrains with an exclusion ("minus") operator. A single exclusion operator may be used without brackets. However when the operands of the exclusion expression constraint are compound, these compound expression constraints must be enclosed by brackets.
<b>BS/LS: dottedExpressionConstraint</b> = subExpressionConstraint 1*(ws dottedExpressionAttribute)	
	A dotted expression constraint contains a sub expression constraint, followed by one or more dotted attributes. When a single dotted attribute is used, the result is the set of attribute values (for the given attribute name) of each concept that results from evaluating the subExpressionConstraint. When more than one dotted attribute is used, each dottedExpressionAttribute is sequentially evaluated (from left to right) against the given result set.

<b>BS/LS: dottedExpressionAttribute</b> = dot ws eclAttributeName	
	A dotted expression attribute consists of a 'dot', followed by an attribute name. Please note that the attribute name may be represented by any sub expression constraint.
<b>BS/LS: subExpressionConstraint</b> = [constraintOperator ws] [memberOf ws] (eclFocusConcept / "(" ws expressionConstraint ws ")")	
	A sub expression constraint optionally begins with a constraint operator and/or a memberOf function. It then includes either a single focus concept or an expression constraint (enclosed in brackets). A memberOf function should be used only when the eclFocusConcept or expressionConstraint refers to a reference set concept, a set of reference set concepts, or a wild card. When both a constraintOperator and a memberOf function are used, they are applied from the inside to out (i.e. from right to left) - see <a href="#">5.4 Operator Precedence</a> . Therefore, if a constraintOperator is followed by a memberOf function, then the memberOf function is processed prior to the constraintOperator.
<b>BS/LS: eclFocusConcept</b> = eclConceptReference / wildCard	
	A focus concept is a concept reference or a wild card.
<b>BS/LS: dot</b> = "."	
	A dot connects an expression constraint with an attribute whose values are included in the result.
<b>BS: memberOf</b> = "^"	
<b>LS: memberOf</b> = "^" / ("m"/"M") ("e"/"E") ("m"/"M") ("b"/"B") ("e"/"E") ("r"/"R") ("o"/"O") ("f"/"F")	
	The 'memberOf' function returns the set of referenced components in the reference set whose concept identifier follows. In the brief syntax, the memberOf function is represented using the "^" symbol. In the long syntax, the text "memberOf " (case insensitive and followed by at least one white space) is also allowed.
<b>BS/LS: eclConceptReference</b> = conceptId [ws " " ws term ws " "]	
	A conceptReference is represented by a ConceptId, optionally followed by a <a href="#">term</a> enclosed by a pair of " " characters. Whitespace before or after the ConceptId is ignored as is any whitespace between the initial " " characters and the first non-whitespace character in the <a href="#">term</a> or between the last non-whitespace character and before second " " character.
<b>BS/LS: conceptId</b> = sctId	
	The ConceptId must be a valid <a href="#">SNOMED CT identifier</a> for a <a href="#">concept</a> . The initial digit may not be zero. The smallest number of digits is six, and the maximum is 18.
<b>BS/LS: term</b> = 1*nonwsnonpipe *( 1*SP 1*nonwsnonpipe )	
	The <a href="#">term</a> must be the <a href="#">term</a> from a <a href="#">SNOMED CT description</a> that is associated with the <a href="#">concept</a> identified by the preceding <a href="#">concept identifier</a> . For example, the <a href="#">term</a> could be the preferred <a href="#">description</a> , or the preferred <a href="#">description</a> associated with a particular translation. The <a href="#">term</a> may include valid <a href="#">UTF-8</a> characters except for the pipe "
<b>BS: wildCard</b> = "***"	
<b>LS: wildCard</b> = "***" / ( ("a"/"A") ("n"/"N") ("y"/"Y"))	
	A wild card represents any concept in the given substrate. In the brief syntax, a wildcard is represented using the "***" symbol. In the long syntax, the text "ANY" (case insensitive) is also allowed.

<b>BS/LS: constraintOperator</b> = childOf / descendantOrSelfOf / descendantOf / parentOf / ancestorOrSelfOf / ancestorOf	
	A constraint operator is either 'childOf', 'descendantOrSelfOf', 'descendantOf', 'parentOf', 'ancestorOrSelfOf', or 'ancestorOf'.
<b>BS: descendantOf</b> = "<"	
<b>LS: descendantOf</b> = "<" / ( ("d"/"D") ("e"/"E") ("s"/"S") ("c"/"C") ("e"/"E") ("n"/"N") ("d"/"D") ("a"/"A") ("n"/"N") ("t"/"T") ("o"/"O") ("f"/"F") mws )	
	The descendantOf operator returns the set of all subtypes of the given concept (or set of concepts). In the brief syntax, the descendantOf operator is represented using the symbol "<". In the long syntax, the text "descendantOf" (case insensitive and followed by at least one white space) is also allowed.
<b>BS: descendantOrSelfOf</b> = "<<"	
<b>LS: descendantOrSelfOf</b> = "<<" / ( ("d"/"D") ("e"/"E") ("s"/"S") ("c"/"C") ("e"/"E") ("n"/"N") ("d"/"D") ("a"/"A") ("n"/"N") ("t"/"T") ("o"/"O") ("r"/"R") ("s"/"S") ("e"/"E") ("l"/"L") ("f"/"F") ("o"/"O") ("f"/"F") mws )	
	The descendantOrSelfOf operator returns the set of all subtypes of the given concept (or set of concepts), plus the concept (or set of concepts) itself. In the brief syntax, the descendantOrSelfOf operator is represented using the symbols "<<". In the long syntax, the text "descendantOrSelfOf" (case insensitive and followed by at least one white space) is also allowed.
<b>BS: childOf</b> = "<!"	
<b>LS: childOf</b> = "<!" / ( ("c"/"C") ("h"/"H") ("i"/"I") ("l"/"L") ("d"/"D") ("o"/"O") ("f"/"F") mws )	
	The childOf operator returns the set of all immediate children of the given concept (or set of concepts). In the brief syntax, the childOf operator is represented using the symbols "<!". In the long syntax, the text "childOf" (case insensitive and followed by at least one white space) is also allowed.
<b>BS: ancestorOf</b> = ">"	
<b>LS: ancestorOf</b> = ">" / ( ("a"/"A") ("n"/"N") ("c"/"C") ("e"/"E") ("s"/"S") ("t"/"T") ("o"/"O") ("r"/"R") ("o"/"O") ("f"/"F") mws )	
	The ancestorOf operator returns the set of all supertypes of the given concept (or set of concepts). In the brief syntax, the ancestorOf operator is represented using the symbol ">". In the long syntax, the text "ancestorOf" (case insensitive and followed by at least one white space) is also allowed.
<b>BS: ancestorOrSelfOf</b> = ">>"	
<b>LS: ancestorOrSelfOf</b> = ">>" / ( ("a"/"A") ("n"/"N") ("c"/"C") ("e"/"E") ("s"/"S") ("t"/"T") ("o"/"O") ("r"/"R") ("o"/"O") ("r"/"R") ("s"/"S") ("e"/"E") ("l"/"L") ("f"/"F") ("o"/"O") ("f"/"F") mws )	
	The ancestorOrSelfOf operator returns the set of all supertypes of the given concept (or set of concepts), plus the concept (or set of concepts) itself. In the brief syntax, the ancestorOrSelfOf operator is represented using the symbols ">>". In the long syntax, the text "ancestorOrSelfOf" (case insensitive and followed by at least one white space) is also allowed.
<b>BS: parentOf</b> = ">!"	
<b>LS: parentOf</b> = ">!" / ( ("p"/"P") ("a"/"A") ("r"/"R") ("e"/"E") ("n"/"N") ("t"/"T") ("o"/"O") ("f"/"F") mws )	
	The parentOf operator returns the set of all immediate parents of the given concept (or set of concepts). In the brief syntax, the parentOf operator is represented using the symbols ">!". In the long syntax, the text "parentOf" (case insensitive and followed by at least one white space) is also allowed.
<b>BS/LS: conjunction</b> = ( ("a"/"A") ("n"/"N") ("d"/"D") mws ) / ", "	

	A conjunction is represented either by the word "and" (case insensitive and followed by at least one white space), or by a comma.
<b>BS/LS: disjunction</b> = ("o"/"O") ("r"/"R") mws	
	A disjunction is represented by the word "or" (case insensitive and followed by at least one white space).
<b>BS/LS: exclusion</b> = ("m"/"M") ("i"/"I") ("n"/"N") ("u"/"U") ("s"/"S") mws	
	The exclusion operator is represented by the word "minus" (case insensitive and followed by at least one white space).
<b>BS/LS: eclRefinement</b> = subRefinement ws [conjunctionRefinementSet / disjunctionRefinementSet]	
	A refinement contains all the grouped and ungrouped attributes that refine the set of clinical meanings satisfied by the expression constraint. Refinements may represent the conjunction or disjunction of two smaller refinements, and may optionally be placed in brackets. Where both conjunction and disjunction are used, brackets are mandatory to disambiguate the intended meaning.
<b>BS/LS: conjunctionRefinementSet</b> = 1*(ws conjunction ws subRefinement)	
	A conjunction refinement set consists of one or more conjunction operators, each followed by a subRefinement.
<b>BS/LS: disjunctionRefinementSet</b> = 1*(ws disjunction ws subRefinement)	
	A disjunction refinement set consists of one or more disjunction operators, each followed by a subRefinement.
<b>BS/LS: subRefinement</b> = eclAttributeSet / eclAttributeGroup / "(" ws eclRefinement ws ")"	
	A subRefinement is either an attribute set, an attribute group or a bracketed refinement.
<b>BS/LS: eclAttributeSet</b> = subAttributeSet ws [conjunctionAttributeSet / disjunctionAttributeSet]	
	An attribute set contains one or more <a href="#">attribute name</a> -value pairs separated by a conjunction or disjunction operator. An attribute set may optionally be placed in brackets.
<b>BS/LS: conjunctionAttributeSet</b> = 1*(ws conjunction ws subAttributeSet)	
	A conjunction attribute set consists of one or more conjunction operators, each followed by a subAttributeSet.
<b>BS/LS: disjunctionAttributeSet</b> = 1*(ws disjunction ws subAttributeSet)	
	A disjunction attribute set consists of one or more disjunction operators, each followed by a subAttributeSet.
<b>BS/LS: subAttributeSet</b> = eclAttribute / "(" ws eclAttributeSet ws ")"	
	A subAttributeSet is either an attribute or a bracketed attribute set.
<b>BS/LS: eclAttributeGroup</b> = [ "[" cardinality "]" ws "{" ws eclAttributeSet ws "]"	
	An <a href="#">attribute group</a> contains a collection of attributes that operate together as part of the <a href="#">refinement</a> of the containing <a href="#">expression</a> constraint. An attribute group may optionally be preceded by a cardinality. An attribute group cardinality indicates the minimum and maximum number of attribute groups that must satisfy the given attributeSet constraint for the expression constraint to be satisfied.
<b>BS/LS: eclAttribute</b> = [ "[" cardinality "]" ws [reverseFlag ws] eclAttributeName ws (expressionComparisonOperator ws subExpressionConstraint / numericComparisonOperator ws "#" numericValue / stringComparisonOperator ws QM stringValue QM )	

	<p>An <b>attribute</b> is a name-value pair expressing a single <b>refinement</b> of the containing <b>expression constraint</b>. Either the attribute value must satisfy (or not) the given expression constraint, the attribute value is compared with a given numeric value (integer or decimal) using a numeric comparison operator, or the attribute value must be equal to (or not equal to) the given string value. The attribute may optionally be preceded by a cardinality constraint and/or a reverse flag.</p>
<p><b>BS/LS: cardinality</b> = minValue to maxValue</p>	
	<p>The cardinality represents a constraint on the minimum and maximum number of times that the given attribute or attribute group may appear in a matching expression. The cardinality is enclosed in square brackets with the minimum cardinality appearing first, followed by a separator (two dots in the brief syntax), and then the maximum cardinality.</p>
<p><b>BS/LS: minValue</b> = nonNegativeIntegerValue</p>	
	<p>A value that represents the minimum number of times that an attribute or attribute group may appear. The minimum cardinality must always be less than or equal to the maximum cardinality.</p>
<p><b>BS: to</b> = ".."  <b>LS: to</b> = ".." / (mws ("t"/"T") ("o"/"O") mws)</p>	
	<p>In the brief syntax, the minimum and maximum cardinality are separated by two dots (i.e. ".."). In the long syntax, the text "to" (case insensitive with at least one white space before and after) is also allowed between the two cardinalities.</p>
<p><b>BS/LS: maxValue</b> = nonNegativeIntegerValue / many</p>	
	<p>A value that represents the maximum number of times that an attribute or attribute group may appear. A maximum cardinality of 'many' indicates that there is no limit on the number of times the attribute may appear.</p>
<p><b>BS: many</b> = "*"  <b>LS: many</b> = "*" / ( ("m"/"M") ("a"/"A") ("n"/"N") ("y"/"Y"))</p>	
	<p>In the brief syntax, a cardinality of 'many' is represented using the symbol "*". In the long syntax, the text "many" (case insensitive, with no trailing space) is also allowed.</p>
<p><b>BS: reverseFlag</b> = "R"  <b>LS: reverseFlag</b> = (("r"/"R") ("e"/"E") ("v"/"V") ("e"/"E") ("r"/"R") ("s"/"S") ("e"/"E") ("o"/"O") ("f"/"F")) / "R"</p>	
	<p>When a reverse flag is used on an attribute, the matching relationships are traversed in the reverse of the normal direction. This means that the target concept of each relationship must match the focus concept to which the attribute is applied, while the source concept of the relationship must match the attribute value. In the brief syntax, the reverse flag is represented using the character "R" (in uppercase). In the long syntax, the text "reverseOf " (case insensitive) is also allowed.</p>
<p><b>BS/LS: eclAttributeName</b> = subExpressionConstraint</p>	
	<p>The attribute name is the name of an attribute (or relationship type) to which a value is applied to refine the meaning of a containing expression constraint. The attribute name is represented using a subExpressionConstraint, as defined above.</p>
<p><b>BS: expressionComparisonOperator</b> = "=" / "!="  <b>LS: expressionComparisonOperator</b> = "=" / "!=" / ("n"/"N") ("o"/"O") ("t"/"T") ws "=" / "&lt;&gt;"</p>	

	Attributes whose value is a concept may be compared to an expression constraint using either equals ("=") or not equals ("!="). In the long syntax "<>" and "not =" (case insensitive) are also valid ways to represent not equals.
<b>BS: numericComparisonOperator</b> = "=" / "!=" / "<=" / "<" / ">=" / ">"	
<b>LS: numericComparisonOperator</b> = "=" / "!=" / ("n"/"N") ("o"/"O") ("t"/"T") ws "=" / "<>" / "<=" / "<" / ">=" / ">"	
	Attributes whose value is numeric (i.e. integer or decimal) may be compared to a specific concrete value using a variety of comparison operators, including equals ("="), less than ("<"), less than or equals ("<="), greater than (">"), greater than or equals (">=") and not equals ("!="). In the long syntax "<>" and "not =" (case insensitive) are also valid ways to represent not equals.
<b>BS: stringComparisonOperator</b> = "=" / "!="	
<b>LS: stringComparisonOperator</b> = "=" / "!=" / ("n"/"N") ("o"/"O") ("t"/"T") ws "=" / "<>"	
	Attributes whose value is numeric may be compared to an expression constraint using either equals ("=") or not equals ("!="). In the long syntax "<>" and "not =" (case insensitive) are also valid ways to represent not equals.
<b>BS/LS: numericValue</b> = ["-"/"+"] (decimalValue / integerValue)	
	A numeric value is either an integer or a decimal. Positive numbers optionally start with a plus sign ("+"), while negative integers begin with a minus sign ("-").
<b>BS/LS: stringValue</b> = 1*(anyNonEscapedChar / escapedChar)	
	A string value includes one or more of any printable ASCII characters enclosed in quotation marks. Quotes and backslash characters within the string must be preceded by the escape character ("\").
<b>BS/LS: integerValue</b> = digitNonZero *digit / zero	
	An integer value is either starts with a non-zero digit followed by zero to many additional digits, or is the integer zero itself.
<b>BS/LS: decimalValue</b> = integerValue "." 1*digit	
	A decimal value starts with an integer. This is followed by a decimal point and one to many digits.
<b>BS/LS: nonNegativeIntegerValue</b> = (digitNonZero *digit ) / zero	
	A non-negative integer value (i.e. positive integers or zero), without a preceding plus sign ("+").
<b>BS/LS: sctId</b> = digitNonZero 5*17( digit )	
	A SNOMED CT id is used to represent an attribute id or a <a href="#">concept id</a> . The initial digit may not be zero. The smallest number of digits is six, and the maximum is 18.
<b>BS/LS: ws</b> = *( SP / HTAB / CR / LF / comment )	

	<p>Optional whitespace characters (space, tab, carriage return, linefeed or a comment) are ignored everywhere in the <code>expression</code> except:</p> <ol style="list-style-type: none"> <li>1. Whitespace within a <code>conceptId</code> is an error. <b>Note:</b> Whitespace before or after the last digit of a valid <code>Identifier</code> is ignored.</li> <li>2. Non-consecutive spaces within a term are treated as a significant character of the term. <b>Note:</b> Whitespace before the first or after the last non-whitespace character of a <code>term</code> is ignored</li> <li>3. Whitespace within the quotation marks of a concrete value is treated as a significant character.</li> </ol>
<b>BS/LS: mws</b> = 1*( SP / HTAB / CR / LF / comment )	
	Mandatory whitespace (i.e. space, tab, carriage return, linefeed or a comment) is required after certain keywords, including "And" and "Or".
<b>BS/LS: comment</b> = "/"*(nonStarChar / starWithNonLSlash) "*"	
	A comment, which provides additional human-readable details about the expression constraint. Comments begin with a forward slash directly followed by a star (i.e. "/"*) and end with a star directly followed by a forward slash (i.e. "*/").
<b>BS/LS: nonStarChar</b> = SP / HTAB / CR / LF / %x21-29 / %x2B-7E / UTF8-2 / UTF8-3 / UTF8-4	
	A character that is not a star (i.e. not %x2A).
<b>BS/LS: starWithNonLSlash</b> = %x2A nonLSlash	
	A star (i.e. "*"*) followed by a character that is not a forward slash (i.e. not "/"").
<b>BS/LS: nonLSlash</b> = SP / HTAB / CR / LF / %x21-2E / %x30-7E / UTF8-2 / UTF8-3 / UTF8-4	
	A character that is not a forward slash (i.e. not "/"").
<b>BS/LS: SP</b> = %x20	
	Space character.
<b>BS/LS: HTAB</b> = %x09	
	Tab character.
<b>BS/LS: CR</b> = %x0D	
	Carriage return character.
<b>BS/LS: LF</b> = %x0A	
	Line feed character.
<b>BS/LS: QM</b> = %x22	
	Quotation mark character.
<b>BS/LS: digit</b> = %x30-39	
	Any digit 0 through 9.
<b>BS/LS: zero</b> = %x30	
	The digit 0.
<b>BS/LS: digitNonZero</b> = %x31-39	

	Digits 1 through 9, but excluding 0.  The first character of a <a href="#">concept identifier</a> is constrained to a digit other than zero.
<b>BS/LS: nonwsnonpipe</b> = %x21-7B / %x7D-7E / UTF8-2 / UTF8-3 / UTF8-4	
	Non whitespace (and non pipe) includes printable ASCII characters (these are also valid UTF8 characters encoded as one octet) and also includes all UTF8 characters encoded as 2- 3- or 4-octet sequences. It excludes space (which is %x20) and the pipe character "
<b>BS/LS: anyNonEscapedChar</b> = SP / HTAB / CR / LF / %x20-21 / %x23-5B / %x5D-7E / UTF8-2 / UTF8-3 / UTF8-4	
	anyNonEscapedChar includes any printable ASCII characters which do not need to be preceded by an escape character (i.e. "\"). This includes valid UTF8 characters encoded as one octet and all UTF8 characters encoded as 2, 3 or 4 octet sequences. It does, however, exclude the quotation mark (") and the backslash (. See RFC 3629 ( <a href="#">UTF-8</a> , a <a href="#">transformation</a> format of <a href="#">ISO 10646</a> authored by the Network Working Group).
<b>BS/LS: escapedChar</b> = BS QM / BS BS	
	The double quotation mark and the back slash character must both be escaped within a string-based concrete value by preceding them with a back slash.
<b>BS/LS: UTF8-2</b> = %xC2-DF UTF8-tail	
	UTF8 characters encoded as 2-octet sequences.
<b>BS/LS: UTF8-3</b> = %xE0 %xA0-BF UTF8-tail / %xE1-EC 2( UTF8-tail ) / %xED %x80-9F UTF8-tail / %xEE-EF 2( UTF8-tail )	
	UTF8 characters encoded as 3-octet sequences.
<b>BS/LS: UTF8-4</b> = %xF0 %x90-BF 2( UTF8-tail ) / %xF1-F3 3( UTF8-tail ) / %xF4 %x80-8F 2( UTF8-tail )	
	UTF8 characters encoded as 4-octet sequences.
<b>BS/LS: UTF8-tail</b> = %x80-BF	
	UTF8 characters encoded as 8-octet sequences.